

DASH Grammar

Jose Serna, Nancy A. Day, and Sabria Farheen
David R. Cheriton School of Computer Science
University of Waterloo
Waterloo, Ontario, Canada, N2L 3G1
Email: {jserna,nday,sfarheen}@uwaterloo.ca

June 16, 2017

```
module ::= (moduleDecl)? import* dashParagraph*
dashParagraph ::= paragraph | topLevelDecl
topLevelDecl ::= 'conc' 'state' name '{' stateDef*}'
stateDef ::= decl | invariantDecl | initDecl | transSet
invariantDecl ::= 'invariant' (name)? block
initDecl ::= 'init' (name)? block
transSet ::= transDecl | stateDecl | eventDecl
           | conditionDecl | actionDecl | addOnDecl
transDecl ::= 'trans' name '{'
             ('from' ('*' | qualNameList))?
             ('on' qualNameList)? ('when' formulas)?
             ('goto' ('*' | qualNameList))? ('do' formulas)?
             ('send' qualNameList)? '}'
stateDecl ::= ('default')? ('conc')? 'state' name '{' stateDef*}'
eventDecl ::= 'event' name transBlock
conditionDecl ::= 'condition' name singleFormula transBlock
actionDecl ::= 'action' name singleFormula transBlock
addOnDecl ::= 'addon' '(' selector formulas ')' 'to' transPattern
selector ::= 'do' | 'send'
transPattern ::= '*' | qualNameList
              | '(' ('from' ('*' | qualNameList))? ('goto' ('*' | qualNameList))? ')'
qualNameList ::= qualName','+
singleFormula ::= '[' formula ']'
formulas ::= formula* | block
block ::= '{' formula*}'
transBlock ::= '{' transSet*}'
```

Terminals in *italic* font have the same definition as in the Alloy grammar
¹. *formula* corresponds to FOL expressions written in Alloy syntax (transitive closure is not supported)

¹The Alloy language documentation: <http://alloy.mit.edu/alloy/documentation.html>